
Oprogramowanie biurowe

OpenOffice.org Calc.

Zaawansowane funkcje finansowe oraz makra.

Ćwiczenie praktyczne, na przykładzie
przygotowania krok-po-kroku arkusza
do estymacji Spłat Kredytu Ratalnego.



Dla kogo jest to ćwiczenie

Jest to dość zaawansowane ćwiczenie. Zakładam, że czytelnik jest już zaznajomiony z podstawową 
obsługą arkusza kalkulacyjnego (czy to MS Excel, czy OO.o Calc) i zamierza  zrobić kolejny krok. 
Krok w kierunku funkcji.

Na początku skorzystamy z kilku gotowych funkcji finansowych. Przećwiczymy jak sprawnie się nimi 
posługiwać.  Przy  okazji,  zobaczymy  też  jak  samemu przeglądać  dostępną  bibliotekę  funkcji  w 
poszukiwaniu właściwych rozwiązań (Kreator funkcji). Na tym etapie zwrócimy szczególną uwagę na 
przyszłą czytelność arkusza (Notatki i Nazwy).

Następnie  zobaczymy  jak  sobie  poradzić,  gdy  biblioteka  nie  zawiera  potrzebnej  nam  funkcji. 
Skorzystamy tu z – jakże wdzięcznego – mechanizmu makr OpenOffice.org Basic.

Ostatecznie wezwiemy na pomoc kilka funkcji  logicznych, aby poprawić przejrzystość gotowego 
arkusza.

Nasze zadanie

Przygotujemy wspólnie arkusz do obliczeń związanych z kredytem ratalnym:

Arkusz podzielony jest na pięć części, w trzech strefach wizualizacji:

I. Definicja podstawowych zmiennych. Oddzielamy w ten sposób wizualnie strefę 
wprowadzania danych (interfejs), od stref wyników.

● Kwota kredytu – ile chcemy pożyczyć pieniędzy,
● Liczba miesięcy – na jak długi okres,
● Roczna stopa procentowa – warunki oferty danego banku.

II. Wyniki poglądowe dla Rat Równych. 

III. Wyniki poglądowe dla Rat Malejących.

● Ile będziemy płacić miesięcznie
● Jaki będzie całkowity koszt takiego kredytu
● Jaką część będziemy mieli spłaconą po zadanym okresie, np. po 1 roku.

IV. Wyniki szczegółowe dla Rat Równych.

V. Wyniki szczegółowe dla Rat Malejących. Lista poszczególnych rat każdego miesiąca.

● Rata kapitałowa – jaką część pożyczonej kwoty oddajemy
● Rata odsetkowa – ile dopłacamy wierzycielowi
● Łączna rata – ile fizycznie musimy zapłacić.

(Strefa pomarańczowa – dane)  (Strefa niebieska – poglądowa)  (Strefa biała – szczegółowa)

Ilustracja 1: Ostateczny wygl d arkusza.ą

Część I – wprowadzenie zmiennych

Ponieważ zakładamy, że podstawowa obsługa arkusza kalkulacyjnego jest nam znana, nie będziemy 
marnować  czasu  na  tworzenie  podstawowego  szkieletu  arkusza.  Skorzystamy  z  wcześniej 
przygotowanego pliku. Otwórz teraz Kredyty-pusty.ods1.

W naszym przykładzie weźmiemy kredyt na 20 000 PLN. W komórce C2 wpisz: 

20000

Jak widzisz, komórka C2 ma już zdefiniowane formatowanie (PPM2 → 'Formatuj komórki'):

● Format walutowy (zakładka 'Liczby' ma 'Kod formatu' = '#,##0.00 [$PLN];[RED]-#,##0.00 
[$PLN]'),

● Czcionka Bold, kolor 'Sea blue' (Lazurowy) – to dobra praktyka poprawiania  czytelności 
dla przyszłych użytkowników arkusza. Zakładamy, że komórki ze zmiennymi (jedyne do 
których  będziemy wprowadzać  dane podczas  przyszłej  pracy  z  arkuszem),  będą pisane 
zawsze jakimś wyróżnionym kolorem.

Kredyt ten będziemy chcieli spłacić w ciągu dwóch lat. W komórce C3 wpisz:

24

Bank zaoferował nam stopę procentową wynoszącą 16,5.  W C4:

16,5

Gdy najedziesz kursorem na komórkę B4 'roczna stopa procentowa' zobaczysz, że dodałem do tej 
komórki  Notatkę (PPM > 'Wstaw notatkę').  Takie  notatki  drastycznie podnoszą czytelności 
złożonych arkuszy. Ta akurat mówi, że...

...Stopa Procentowa którą oferują man banki jest  Stopą Procentową w Skali Roku, my do naszych 
obliczeń  będziemy  potrzebowali  Stopy  Procentowej  w  Skali  Okresu  Rozliczeniowego, w  tym 
przypadku Miesiąca. W C5:

=C4/12

Tekst w komórce C5 (Miesięczna Stopa Procentowa) nie ma już specjalnego formatowania, bo nigdy 
nie  będzie  bezpośrednio  edytowany  przez  użytkownika  arkusza.  Będzie  zawsze  obliczany  na 
podstawie Rocznej Stopy Procentowej.

Jeśli  wszystko się udało to powinniśmy uzyskać efekt podobny do tego na Ilustracji  .  Wszystkie 
nasze zmienne zgrupowaliśmy w jednym miejscu. Przejdźmy teraz do obliczeń na tych zmiennych.

1 Podczas otwierania plików zawierających makra, Calc może pytać czy włączyć ich obsługę.
2 PPM – Prawy Przycisk Myszy ma danej komórce.

Ilustracja  2:  Zwi kszamy  czytelno ć:  Komórki  zeę ś  
zmiennymi  s  graficzne  wyró nione.  Notatka  wą ż  
komórce  B4  jest  zasygnalizowana  czerwonym 
kwadracikiem w jej prawym górnym rogu.



Część II – wyniki poglądowe dla Rat Równych

Raty Równe to takie gdzie co miesiąc wpłacamy stałą kwotę raty łącznej3. Różnią się one pod tym 
względem od  Rat  Malejących,  przy  których  co  miesiąc  spłacamy  stałą  kwotę  raty  kapitałowej, 
łączna rata jest natomiast zmienna. W praktyce przy Ratach Równych, co miesiąc wpłacamy taką 
samą kwotę, przy Malejących co miesiąc troszkę mniej.

Łączna miesięczna rata

Jeśli co miesiąc płacimy tyle samo, to jaka jest ta kwota? A no wynosi ona tyle:

RataRówna=
KwotaKredytu⋅MiesięcznaStopaProc⋅

1−1MiesięcznaStopaProc⋅ −LiczbaMiesięcy

Wzór  ten  nie  jest  bardzo  skomplikowany  i  jeśli  mamy  wprawę  w  tworzeniu  formuł  możemy 
szybciutko w C10 napisać:

=C2*C5%/(1-(1+C5%)^-C3)

Jednak elegantszym, rozwiązaniem jest skorzystanie z odpowiedniej funkcji. Funkcją taką jest PMT() 
(od ang. PayMenT – płatność, spłata). Na stronie pomocy (F1 → Arkusze kalkulacyjne → Operatory i 
typy funkcji → Funkcje finansowe – część druga) znajdujemy informację:

PMT()

Zwraca płatność okresową dla rat rocznych przy założeniu stałej stopy procentowej.

PMT(stopa ; NPER; PV; FV; typ)

Stopa: okresowa stopa procentowa.

NPER: liczba okresów płatności rat rocznych.

PV: bieżąca wartość gotówkowa w sekwencji płatności.

FV  (parametr  opcjonalny):  wartość  żądana  (wartość  przyszła)  osiągana  na  koniec 
płatności okresowych.

Typ (parametr opcjonalny): data płatności okresowych. Typ = 1 oznacza płatność na  
początku każdego okresu, a Typ = 0 oznacza płatność na końcu każdego okresu.

Praca z Kreatorem funkcji

Skorzystajmy z tej funkcji za pomocą kreatora funkcji. Upewnij się, że masz wybraną komórkę C10 i 
kliknij  na ikonę   na  pasku formuły lub wybierz z menu 'Wstaw'  → 'Funkcja... (Ctrl + F2)'. 
Wybierz kategorie 'Finanse'. Odszukaj PMT na liście funkcji i kliknij 'Dalej >>'.

3 Rata Łączna = Rata Kapitałowa (część spłacana) + Rata Odsetkowa (suma dopłacana)

Gdy  przechodzimy  pomiędzy  polami  parametrów  funkcji,  kreator  podpowiada  nam  jakie  jest 
znaczenie danego parametru. I  choć robi  to nie najprecyzyjniej  w tym konkretnym przypadku – 
funkcja ma parę zastosowań – to wystarczająco by się domyśleć. I tak:

● Stopa to to samo co nasza 'miesięczna stopa procentowa' tyle że wyrażona w 
procentach, czyli C5%.

● NPER to nasza 'liczba miesięcy', czyli C3.
● PV to nasza 'kwota kredytu' czyli C2.

Klikamy 'OK' i już w komórce C10 mamy taki kod formuły:

=PMT(C5%;C3;C2)

Kod ten jest czytelniejszy od kodu całego wzoru, który napisaliśmy wcześniej. Wystarczy, że znamy 
funkcję PMT, i od razu widzimy co liczymy. Czytelność można jednak jeszcze zwiększyć.

Definicje Nazw

Odwoływanie  się  do  często  używanej  i 
posiadającej  logiczne  znaczenie  zmiennej 
poprzez  adres  komórki  np.  C2  nie  jest 
eleganckim  rozwiązaniem.  Znacznie 
zwiększymy  czytelność  formuł  jeśli 
wprowadzimy Nazwy dla naszych zmiennych.

Wybierz  komórkę  C2,  a  następnie  z  menu 
'Wstaw' → 'Nazwy' → 'Definiuj... Ctrl+F3'.

Od razu zacznij  pisać 'KwotaKred'.  W dolnej 
części  'Przypisano  do'  upewnij  się,  że 
pracujesz  na komórce C2.  Następnie  wybierz 
'Dodaj'.

Od razu zacznij  pisać 'LiczbaMies'.  W dolnej 
części  'Przypisano  do'  kliknij  ikonę 
'Pomniejsz'   i  wybierz LPM4 komórkę  C3. 
Jeszcze raz 'Pomniejsz' i 'Dodaj'.

Tak samo stwórz 'StopaProcMies' dla C5.

Zakończ wprowadzanie nazw klikając 'OK'.

4 LPM – Lewy Przycisk Myszy na komórce.

Ilustracja  4:  Najpierw  zacznij  pisać  now  nazw ,ą ę  
upewnij si  gdzie przypisujesz i kliknij 'Dodaj'.ę

Ilustracja 5: Po 'Dodaniu', zacznij pisać now  nazw .ą ę

Ilustracja 3: Ka dy parametr posiada tekstow  podpowiedz jego znaczenia.ż ą



Wybierz teraz komórkę C2 i zwróć uwagę, że lista 
rozwijana  o  nazwie  Obszar  arkusza na  pasku 
formuły pokazuje teraz, zamiast adresu komórki 
'C2',  Nazwę zmiennej  KwotaKred zawartej w tej 
komórce. Rozwijając listę  Obszar arkusza mamy 
dostęp  do  wszystkich  Nazw zdefiniowanych  w 
arkuszu.

Wprowadzanie dynamiczne

Gdy mamy już zdefiniowane Nazwy dla naszych zmiennych, przejdźmy jeszcze raz do komórki C10. 
Skasujmy jej stary kod (Del na klawiaturze) i wpiszmy nowy kod z użyciem naszych zmiennych.

Ponieważ znamy już składnie funkcji PMT nie będziemy korzystać z Kreatora funkcji tylko zaczniemy 
od razu pisać:

=P

Już teraz Calc dynamicznie podpowie nam, czy chcemy użyć funkcji PMT().

Potwierdźmy naciskając Enter na klawiaturze. Piszmy dalej:

St

Calc dynamicznie podpowie nam czy chcemy użyć StopaProcMies.

Zatwierdźmy Enterem. Piszmy dalej:

%;L

I znowu podpowiedz. Czy chodzi nam o LiczbaMies? Enter i dalej:

;K

KwotaKred? Enter. I już mamy całość wprowadzoną. Zakończmy edycje formuły naciskając jeszcze 
raz Enter.

Wykorzystując Nazwy byliśmy wstanie bardzo szybko, nie odrywając rąk od klawiatury, bez żadnego 
klikania  myszką czy błądzenia  po  arkuszu,  wprowadzić  odpowiednią formułę.  I  to  jaką formułę! 
Porównajmy na koniec jej czytelność z naszymi poprzednimi próbami:

=C2*C5%/(1-(1+C5%)^-C3)

=PMT(C5%;C3;C2)

=PMT(StopaProcMies%;LiczbaMies;KwotaKred)

Ilustracja  6: Lista rozwijana Obszar arkusza na pasku 
formuły ułatwia dost p do zdefiniowanych ę Nazw.



Suma rat odsetkowych (koszt kredytu)

Drugą istotą informacją dla porównania różnych wariantów kredytu, jest obliczenie ile zapłacimy za 
wszystkie odsetki razem. To znaczy, ile musimy dopłacić do tego interesu. Algebraicznie ujęlibyśmy 
to tak:

SumaOdsetek=PMT∗LiczbaMiesięcy−KwotaKredytu

Jest jednak specjalna funkcja, pozwalająca sumować odsetki z danych przedziałów płatności. Np. 
odsetki  od  3  do  6  miesiąca.  W szczególnym przypadku  (od  początku  do końca)  zsumuje  nam 
wszystkie odsetki. Taką funkcją jest CUMIPMT (od ang. Cumulative Interest – kumulujące, sumujące 
się odsetki).

CUMIPMT()

Oblicza łączne odsetki wypłacane dla inwestycji o stałej stopie procentowej.

CUMIPMT(stopa; NPER; PV; okres_początkowy; okres_końcowy; typ)

Stopa: okresowa stopa procentowa.

NPER: okres płatności obejmujący łączną liczbę okresów. Wartość NPER nie musi być  
liczbą całkowitą.

PV: wartość bieżąca w sekwencji płatności.

Okres_początkowy: pierwszy okres.

Okres_końcowy: ostatni okres.

Typ: data płatności przypadająca na początek lub koniec okresu.

Jak widzimy składnia jest podobna do składni funkcji PMT. Dodatkowe są parametry

● początku i końca okresu – dla nas pierwszy (1) i ostatni (24) miesiąc
● typ, który przyjmie u nas wartość 0 – płatność na końcu miesiąca.

Wprowadź teraz, dynamicznie lub korzystając z Kreatora funkcji odpowiedni kod w komórce C11:

=CUMIPMT(StopaProcMies%;LiczbaMies;KwotaKred;1;LiczbaMies;0)

Wybór  pomiędzy  kreatorem  funkcji  a  wprowadzaniem  dynamicznym  bywa  czasem  kłopotliwy. 
Kreator ma tę zaletę, że wyświetla przez cały czas tekst pomocy. Jeśli  nie pamiętasz kolejności 
parametrów danej funkcji  Kreator Ci pomoże, jednak nie podpowie podczas wprowadzania  Nazw 
zmiennych. Te trzeba będzie wpisać w całości ręcznie. Jeśli użyjesz długich, czytelnych nazw, może 
być to kłopotliwe. Jeśli użyjesz zbyt krótkich nazw, utracisz późniejszą czytelność formuły.

Ilustracja 7: Kreator funkcji nie uzupełnia dynamicznie Nazwy zmiennej. Trzeba 
wpisać j  w cało ci r cznie.ą ś ę



Suma rat kapitałowych

Istotną informacją, szczególnie dla podkreślenia różnic między Rotami Równym a Malejącymi jest 
sprawdzenie ile pieniędzy oddamy bankowi po zadanym okresie. Lub ogólniej, jaka jest suma rat 
kapitałowych w danym przedziale płatności.

W naszym przykładzie okresem tym będzie pierwszy rok, lecz arkusz nasz chcemy, aby potrafił 
obliczyć tę informację dla dowolnego okresu. Stwórzmy zatem dwa pola, dzięki  którym przyszły 
użytkownik będzie mógł dostosować obliczenia do swoich potrzeb.

W komórce C12 wpisz:

1

W C13:

12

Ponieważ,  są  to  pola  zmienne,  pola  'interfejsu'  przyszłego  użytkownika,  używamy  dla  nich 
specjalnego koloru czcionki.

Od obliczeń posłużymy się natomiast funkcją CUMPRINC (ang. Cumulative Principal – kumulujący, 
sumujący się kapitał):

CUMPRINC()

Zwraca łączne wypłacane odsetki dla okresu inwestycji o stałej stopie procentowej.

CUMPRINC(stopa; NPER; PV; okres_początkowy; okres_końcowy; typ)

Stopa: okresowa stopa procentowa.

NPER: okres płatności obejmujący łączną liczbę okresów. Wartość NPER nie musi być  
liczbą całkowitą.

PV: wartość bieżąca w sekwencji płatności.

Okres_początkowy: pierwszy okres.

Okres_końcowy: ostatni okres.

Typ: data płatności przypadająca na początek lub koniec okresu.

Funkcja CUMPRINC jest odpowiednikiem CUMIPMT dla rat kapitałowych. Nic więc dziwnego, że jest 
składnia jest taka sama.

W C14 wprowadź:

=CUMPRINC(StopaProcMies%;LiczbaMies;KwotaKred;C12;C13;0)

Część III – wyniki poglądowe dla Rat Malejących

Łączna miesięczna rata

Raty Malejące mają malejącą Łączną Ratę wyliczoną na podstawie stałej Raty Kapitałowej i odsetek 
z niespłaconej jeszcze części długu:

ŁącznaRataMalejacan=RataKapitałowaPozostałyDługn⋅StopaProcentowa

gdzie

PozostałyDługn=KwotaKredytu−RataKapitalowa⋅n−1

Ponieważ  wartość  raty  jest  zmienna,  w części  poglądowej  naszego arkusza  nie  wystarczy  nam 
pojedyncza wartość jak w przypadku Rat Równych. Ale, ponieważ łatwo dostrzec, że nasza Rata 
Równa będzie maleć liniowo, wystarczy użycie dwóch komórek, pewnego zakresu, od – do. Przyszły 
użytkownik łatwo będzie mógł sobie wyobrazić, że wszytko pomiędzy to liniowa interpolacja.

Szczególnie informacja o pierwszej, najwyższej racie, będzie dla użytkownika istotna, bo banki na 
podstawie tzw. Zdolności Kredytowej Klienta obliczają, jaką mogą zaoferować Maksymalną Kwotę 
Miesięcznej Spłaty. Jeśli wartość pierwszej raty przekroczy tę wartość, bank nie udzieli nam kredytu.

Niestety w puli funkcji finansowych nie znajdziemy odpowiednika funkcji PMT dla rat malejących. 
Powyższy wzór jest jednak na tyle skomplikowany, że niewygodnie byłoby go używać bezpośrednio 
w formule. Np. dla pierwszej raty:

=KwotaKred/LiczbaMies+(KwotaKred-
(KwotaKred/LiczbaMies*(1-1)))*StopaProcMies%

Stwórzmy zatem, przy pomocy makr własną funkcję, analogiczną do PMT. Nazwiemy ją DECPMT (od 
Decreasing Payment). Jej prototyp natomiast będzie prawie identyczny do PMT:

DECPMT(stopa ; NPER; PV; Okres)

Różnicą  jest  dodatkowy  parametr  Okres,  dzięki  któremu  przekazywać  będziemy  odpowiedni 
miesiąc.

Wybierz z menu 'Narzędzia' → 'Makra' → 'Zarządzaj makrami' → 'Makro'.

Na liście 'Makro z' wybierz plik 'Kredyty-pusty', dzięki czemu nowe makro zostanie zapisane razem z 
naszym plikiem. Następnie wybierz 'Nowy'.  Nasz plik nie ma jeszcze zdefiniowanego Modułu do 
przechowywania  makr,  więc  Calc  zapytana  o  nazwę  dla  nowego  modułu.  Domyślna  nazwa 
'Module1' jest ok, czyli kliknij 'OK'.

Ilustracja 8: Makra przechowywane s  w Modułach.ą



Zobaczysz teraz okno OpenOffice.org Basic z nową, pustą dyrektywą Main. Kod naszej nowej funkcji 
wprowadzimy przed nią, zaraz za linijką 'REM  *****  BASIC  *****':

Function DECPMT(Stopa, NPER, PV, Okres)

rem Rata łączna dla Rat Malejących. 

rem Stopa: okresowa stopa procentowa.

rem NPER: liczba okresów płatności rat rocznych.

rem PV: kwota kredytu

rem Okres: okres płatności. Okres=1 dla pierwszego, 
Okres=NPER dla ostatniego.

RataKapitalowa = PV / NPER

PozostalyDlug = PV - (RataKapitalowa * (Okres - 1))

DECPMT = RataKapitalowa + PozostalyDlug * Stopa

DECPMT = -DECPMT

End Function

Jest to naprawdę prosty kod. Jeśli miałeś wcześniej jakąkolwiek styczność z jakimkolwiek językiem 
programowania,  to  pewnie  wszytko jest  jasne.  Jeśli  nie  miałeś,  to  może warto  dodać,  że  linijki 
zaczynające się od 'rem', to linijki komentarza – teoretycznie niepotrzebne, praktycznie niezbędne 
dla  przyszłej  czytelności.  Pierwsza  i  ostatnia  linijka,  zaczynają  i  kończą  definicje  nowej  funkcji. 
Reszta, to te same wzory, które wprowadziliśmy sobie na początku rozdziału, zapisane zgodnie z 
deklaracją zmiennych w OO.o Basicu.  Widzimy też,  że aby zdefiniować wartość zwracaną przez 
funkcję, wystarczy użyć nazwy funkcji, tu 'DECPMT =...'. Na koniec odwracamy jeszcze znak funkcji, 
aby zwiększyć podobieństwo do PMT.

I  tyle.  Stworzyliśmy właśnie  nową funkcję.  Zamknij  teraz  okno  OO.o  Basic,  i  przekonaj  się,  że 
możesz już używać naszej nowe funkcji, jak każdej innej. Zaczniemy od pierwszej raty. Przejdź do 
C18 i wprowadź:

=DECPMT(StopaProcMies%;LiczbaMies;KwotaKred;1)

Następnie, obliczymy ostatnią ratę w C19:

=DECPMT(StopaProcMies%;LiczbaMies;KwotaKred;LiczbaMies)

Suma rat odsetkowych

Nie ma co się oszukiwać, odpowiednika funkcji CUMIPMT dla rat malejących, też nie znajdziemy w 
puli  gotowych  funkcji.  Pójdziemy  jednak  za  ciosem  i  ponownie  stworzymy  sobie  odpowiednią 
funkcję. Nazwa będzie logiczną kontynuacją poprzednich nazw: CUMIDECPMT (Cumulative Interest 
Decreasing Payment), a prototyp prawie identyczny z CUMIPMT:

CUMIDECPMT(Stopa, NPER, PV, OkresPoczatek, OkresKoniec)

Jednak tutaj zadanie nie będzie już tak trywialne ponieważ będziemy musieli użyć sumy:

∑
i=OkresPoczątek

OkresKoniec

PozostałyDługi⋅StopaProcentowa

Gdzie pozostały dług podobnie jak poprzednio:

PozostałyDługi=KwotaKredytu−RataKapitalowa⋅i−1

Sumę tę zapiszemy w OO.o Basicu przy pomocy pętli.  Otwórz ponownie okno edycji  OO.o Basic 
('Narzędzia'  → 'Makra'  → 'Zarządzaj  makrami'  → 'Makro';  znajdź  'Kredyty-pusty'  → 'Standard'  → 
'Module1' i kliknij 'Edytuj'). Zaraz za definicją funkcji DECMPT, czyli pomiędzy linijką 'End Function', 
a 'Sub Main' wprowadź:

Function CUMIDECPMT(Stopa,NPER,PV,OkresPoczatek,OkresKoniec)

rem Suma rat odsetkowych dla Rat Malejących.

rem Stopa: okresowa stopa procentowa.

rem NPER: liczba okresów płatności rat rocznych.

rem PV: kwota kredytu

rem OkresPoczatek: piewszy Okres płatności.

rem OkresKoniec: ostatni Okres.

RataKapitalowa = PV / NPER

CUMIDECPMT = 0

for i = OkresPoczatek to OkresKoniec

PozostalyDlug = PV - (RataKapitalowa * (i - 1))

CUMIDECPMT = CUMIDECPMT + PozostalyDlug * Stopa

next i

CUMIDECPMT = -CUMIDECPMT

End Function

Całość  jest  analogiczna  do  naszej  poprzedniej  funkcji  DECPMT.  Do  obliczenia  sumy  używamy 
natomiast pętli 'For... Next'5. Ida jest taka, że najpierw upewniamy się że zaczynamy od zera. Potem 
pomiędzy linijkami 'For...' i 'Next' obliczamy ratę odsetkową dla pierwszego okresu i dodajemy ją do 
naszego, zera. Potem obliczamy dla drugiego kresu i dodajemy do poprzedniego wyniku i tak dalej, 
aż do ostatniego okresu. Na koniec uzbiera nam się cała Suma.

Zamknij okno OO.o Basic i przekonaj się, jak działa nowa funkcja obliczając sumę rat odsetkowych 
w C20:

 =CUMIDECPMT(StopaProcMies%;LiczbaMies;KwotaKred;1;LiczbaMies)

5 Więcej na temat pętli znajdziesz w 'Pomoc' → 'Pomoc OpenOffice.org F1' → Makra i 
programowanie' → 'Lista poleceń' → 'Uruchamianie funkcje, polecenia i operatory' → Sterowanie 
wykonywaniem programu' → 'Pętle'

Suma rat kapitałowych

Jak i wcześniej sprawdzimy jaką część kapitału będziemy mieli spłaconą po pierwszym roku. W C21:

1

W C22:

12

Rata kapitałowa w przypadku rat malejących jest stała, więc obliczenie jej sumy jest jest naprawdę 
proste, nie będą nam do tego potrzebne żadne makra ani specjalne funkcje. W C23:

=-(KwotaKred/LiczbaMies)*(C22-C21+1)

Minus  na  początku  jest  nam  potrzebny,  aby  upodobnić  wynik  do  wyniku  poprzednio  użytej 
CUMPRINC.

Część IV – wyniki szczegółowe dla Rat Równych

Stworzymy  teraz,  łatwą  do  drukowania  listę  poszczególnych  rat  każdego  miesiąca.  Pierwsza 
kolumna 'numer raty' będzie naszym licznikiem, funkcje których użyjemy będą go potrzebowały. 
Dodatkowo chcielibyśmy, aby lista ta działała prawidłowo dla dłuższych, np. 3 letnich, kredytów.

Rata kapitałowa

Ratę kapitałową umielibyśmy już policzyć korzystając z CUMPRINC. Posłużmy się jednak właściwą 
funkcją:

PPMT()

Oblicza  wartość  spłaty  kapitału  dla  podanego  okresu  inwestycji  przy  założeniu  
regularnych stałych wpłat i stałej stopy oprocentowania.

PPMT(stopa; okres; NPER; PV; FV; typ)

Stopa: okresowa stopa procentowa.

Okres: okres amortyzacji. P = 1 dla pierwszego i P = NPER dla ostatniego okresu.

NPER: całkowita liczba okresów płatności rat rocznych.

PV. wartość bieżąca w sekwencji płatności.

FV (parametr opcjonalny): wartość żądana (przyszła).

Typ (parametr opcjonalny): definiuje datę płatności. F = 1 dla płatności na początku 
okresu; F = 0 dla płatności na koniec okresu.

Wpisz w F4:

=-PPMT(StopaProcMies%;E4;LiczbaMies;KwotaKred)

Naszym  licznikiem  Okresu  jest  kolumna  E,  więc  dla  F4  użyjemy  E4,  dla  F5  użyjemy  E5  itd. 
Oczywiście nie będziemy tego robić ręcznie, tylko rozciągniemy naszą formułę na niższe komórki. 

Upewnij się, że masz wybraną komórkę F4. Naciśnij LPM6 na czarny kwadracik w prawym dolnym 
rogu tej komórki (kursor mysz zmieni się na krzyżyk), przeciągnij mysz w dół, aż do komórki F39, 
puść LPM.

Obliczenia dla okresów poniżej 24 miesiąca dadzą na razie komunikat 'Błąd:504'. Poradzimy sobie z 
tym później.

6 Lewy Przycisk Myszy

Rata odsetkowa

Identyczna jak poprzednia jest funkcja:

IPMT()

Oblicza amortyzację okresową dla inwestycji  przy założeniu regularnych płatności  i  
stałej stopy oprocentowania.

IPMT(stopa; okres; NPER; PV; FV; typ)

Stopa oznacza okresową stopę procentową.

Okres definiuje okres obliczeniowy odsetek składowych. W celu obliczenia odsetek 
składowych dla ostatniego okresu należy przyjąć, że okres = NPER.

NPER jest całkowitą liczbą okresów płatności rat rocznych.

PV oznacza bieżącą wartość gotówkową w sekwencji płatności.

FV  (parametr  opcjonalny)  definiuje  wartość  żądaną  (wartość  przyszłą)  na  koniec  
okresów rozliczeniowych.

Typ oznacza datę płatności okresowych.

Tak samo jak dla raty kapitałowej, w G4:

=-IPMT(StopaProcMies%;E4;LiczbaMies;KwotaKred)

A następnie rozciągnij tę formułę w dół, aż do G39.

Rata łączna

Z funkcją PMT jesteśmy już zaprzyjaźnieni. W H4:

=-PMT(StopaProcMies%;LiczbaMies;KwotaKred)

A potem rozciągnij do H39.

Minus na początku wszystkich formuł tego rozdziału użyliśmy, aby poprawić wygląd wydruków.

Część V – wyniki szczegółowe dla Rat Malejących

Rata kapitałowa

Jak i poprzednio, obejdziemy się bez funkcji. W I4:

=KwotaKred/LiczbaMies

I rozciągnij do I39.

Rata odsetkowa

Posłużymy się naszym makrem do liczenia Sumy Rat Odsetkowych. Suma z jednego okresu, da 
wartość tego okresu. W J4:

=-CUMIDECPMT(StopaProcMies/100;LiczbaMies;KwotaKred;E4;E4)

Rozciągnij do J39.

Rata łączna

Makro do liczenia raty łącznej, też mamy już gotowe. W K4:

=-DECPMT(StopaProcMies/100;LiczbaMies;KwotaKred;E4)

Rozciągnij do K39.

Funkcje logiczne – walidacja zakresów

Biała tabela, szczegółowej rozpiski rat, nie działa jednak dobrze. Obliczenia są w porządku, ale jest 
ona troszkę za mało dynamiczna. Już teraz pokazuje błędy po 24 miesiącu. Jednak ideą naszego 
arkusza było, aby liczba miesięcy mogła być dowolnie modyfikowana. Dobrze by było, gdyby nasze 
formuły sprawdzały jaka jest aktualnie zdefiniowana liczba miesięcy i nie wypisywały 'bzdur'.

Nic prostszego. Pomoże nam funkcja:

JEŻELI()

Wykonuje test logiczny.

JEŻELI(Test; Wtedy_wartość; W_przeciwnym_razie_wartość)

Test jest wyrażeniem mogącym przybierać wartość PRAWDA lub FAŁSZ.

Wtedy_wartość (opcjonalnie) jest wartością zwracaną przez funkcję jeśli Test zwraca 
PRAWDA.

W_przeciwnym_razie_wartość (opcjonalnie) jest wartością zwracaną przez funkcję jeśli  
Test zwraca FAŁSZ.

W istocie,  używamy jej  tutaj  głównie po to,  aby zobaczyć jak można zagnieżdżać wyniki  jednej 
funkcji jako argumenty dla drugiej.

Spróbujmy poprawić formułę w F4 tak, że gdy nasz licznik (kolumna E) przekroczy liczbę miesięcy, 
zamiast wyniku funkcji PPMT, zobaczymy napis '---'.

=JE ELI(E4>LiczbaMies;"---";-Ż
PPMT(StopaProcMies/100;E4;LiczbaMies;KwotaKred))

Popraw tak samo komórki G4-K4. Następnie zaznacz cały rząd od F4 do K4 i rozciągnij formuły w 
dół, aż do F39-K39.

Teraz, niezależnie na ile miesięcy chcemy wziąć nasz kredyt, lista poszczególnych rat zawsze poda 
sensowny  wynik.  Jeśli  natomiast  będziemy chcieli  wziąć  kredyt  na  dłużej  niż  trzy  lata,  zawsze 
możemy wydłużyć naszą listę w dół, dokładnie jak przed chwilą, rozciągając zakres formuł w dół... 
tak nisko, jak sobie zażyczymy.

Na koniec, możesz porównać efekty własnej pracy z plikiem Kredyty-wersja.ostateczna.ods.

Ilustracja 9: Rozci ganie zakresu formuły przy kilku komórkach zaznaczonych.ą



Podsumowanie

Gdy mamy już gotowy arkusz, łatwo dostrzec pewne cechy szczególne Rat Równych i Malejących. 
Jak widzisz, kredyt spłacany w Ratach Równych, będą zawsze droższy. Z drugiej strony Maksymalna 
Miesięczna Spłata będzie zawsze wyższa przy Malejących. Zatem, choć oczywiście bardziej opłaca 
się  spłacać  kredyt  w  ratach  Malejących,  wybierając  raty  Równe  podnosimy  naszą  Zdolność 
Kredytową, dzięki czemu możemy wziąć wyższy kredyt, na krótszy okres.

Użyte przez nas wbudowane funkcje finansowe:

PMT (PayMenT) rata łączna

PPMT (Principal Payment) rata kapitałowa

IPMT (Interest Payment) rata odsetkowa

CUMPRINC (Cumulative Principal) suma rat kapitałowych w danym okresie

CUMIPMT (Cumulative Interest) suma rat odsetkowych w danym okresie

16/16


	Dla kogo jest to ćwiczenie
	Nasze zadanie
	Część I – wprowadzenie zmiennych
	Część II – wyniki poglądowe dla Rat Równych
	Łączna miesięczna rata
	PMT()
	Praca z Kreatorem funkcji
	Definicje Nazw
	Wprowadzanie dynamiczne

	Suma rat odsetkowych (koszt kredytu)
	CUMIPMT()

	Suma rat kapitałowych
	CUMPRINC()


	Część III – wyniki poglądowe dla Rat Malejących
	Łączna miesięczna rata
	Suma rat odsetkowych
	Suma rat kapitałowych

	Część IV – wyniki szczegółowe dla Rat Równych
	Rata kapitałowa
	PPMT()

	Rata odsetkowa
	IPMT()

	Rata łączna

	Część V – wyniki szczegółowe dla Rat Malejących
	Rata kapitałowa
	Rata odsetkowa
	Rata łączna

	Funkcje logiczne – walidacja zakresów
	JEŻELI()

	Podsumowanie

